

DEPARTMENT OF THE INTERIOR

U.S. GEOLOGICAL SURVEY

PORTING THE EXCITATION-EMISSION-MATRIX 3-D PLOTTING PROGRAM
FROM APPLESOFT BASIC TO IBM COMPATIBLE GW-BASIC

Arnold F. Theisen

U.S. Geological Survey

Denver Federal Center, MS-964

Denver, Colorado 80225

Open-File Report 89-312

This report is preliminary and has not been reviewed for conformity with U.S. Geological Survey editorial standards. Any use of trade names is for descriptive purposes only and does not imply endorsement by the USGS.

PORTING THE EXCITATION-EMISSION-MATRIX 3-D PLOTTING PROGRAM
FROM APPLESOFT BASIC TO IBM COMPATIBLE GW-BASIC

Excitation-emission-matrix (EEM) room temperature fluorescence has been a major tool for the study of the fluorescence properties of materials by members of the Luminescence Studies Project of the U.S. Geological Survey. The display of EEM data as a three dimensional perspective plot has been a key factor in the successful interpretation of these data. The in-house developed software to produce these 3-D plots has been hardware specific until now.

EEM fluorescence data generated with a Perkin-Elmer MPF-series spectrometer was collected and stored using a Cyborg Corporation Integrated System for Automated Acquisition and Control (ISAAC). The ISAAC owned by this project, uses an Apple II for a central processing unit (CPU). Although other ISAAC-CPU combinations are now available, this was not originally the case. For details of this setup see Theisen and Hemphill, 1985.

The 3-D perspective software has evolved from Apple II machine language and a program called A2-3D2 from SubLogic to an AppleSoft BASIC program using a simple algorithm obtained from "Microcomputer Graphics" (Myers, 1982). Included in the same book is a hidden-line removal algorithm, a feature the A2-3D2 package did not provide.

With any program that requires extensive calculations, as this one does, speed is an absolute necessity. To achieve an acceptable level of speed on the Apple II, a 10 MHz Motorola MC68000-based coprocessor board (PDQ II) from Enhancement Technology was purchased and installed. Data for the mineral Barite was collected and plotted (figure 1) in approximately one hour.

Recent collaborative efforts between this project and the Remote Sensing group from the U.S. Army Engineer Topographic Laboratories (ETL) has created a requirement, by ETL, to make the same kinds of measurements using an SLM-Aminco fluorescence spectrometer. The CPU for the SLM is an IBM-PC and all the microcomputers the ETL Remote Sensing group uses are either IBM-PC's or PC clones. The 3-D plotting software has, therefore, been ported to IBM-PC compatible GW-BASIC.

The original Apple II program (listing I) reads previously stored binary format EEM data, transforms 3-D data into a 2-D representation, and checks for hidden lines as the information is being plotted. Listing II is the GW-BASIC port of this program developed on a GRID-3 laptop microcomputer. While the Apple II version was coupled to a Houston Instrument (HI) plotter and therefore used HI plotter commands, the GW-BASIC version is set up for Hewlett-Packard (HP) and compatible plotters and uses HP Graphics Language (HPGL). The other major difference with the GW-BASIC program is that it reads previously stored ASCII format EEM data. For both listings, more than two spaces in a print statement are numbered inside brackets (e.g. [3] represents 3 spaces).

REFERENCES

Myers, R.E., 1982, Microcomputer Graphics: Addison-Wesley Publishing Company, Reading, Mass., 282 pages.

Theisen, A.F. and Hemphill, W.R., 1985, Microcomputers in the luminescence laboratory: A technique for automating spectrometers: American Laboratory, v. 17, no. 9, p. 166-171.




Figure 1. Excitation-emission-matrix plot example, the mineral Barite.

Produced with an Enhancement Technology's PDQ II equipped Apple II
on a Houston Instrument DMP-29.

Listing I (EEM 3-D PLOT)

```

1    REM CHANGED 4/14/89

10   GOTO 1000

17   REM *****

18   REM 3 - D to 2 - D translation routine

19   REM *****

20   TX = - PX * S1 + PY * C1

30   TY = - PX * C1 * C2 - PY * S1 * C2 + PZ * S2

40   TZ = - PX * S2 * C1 - PY * S2 * S1 - PZ * C2 + RHO

50   SX = (D * TX / TZ + 36.6050347) * 24.7680682 + 160

60   SY = (D * TY / TZ + 25.2546018) * 23.4965782 + 1

70   RETURN

77   REM *****

78   REM Houston Instrument line segment plot command

79   REM *****

80   SG$ = ";:A" + STR$ (X + 280) + "," + STR$ (Y + 75)

90   PRINT SG$:
    SG$ = "":
    SG = FRE (0)

100  RETURN

107  REM *****

108  REM Hidden line removal routine

109  REM *****

110  IF Y < = MAX%#(X) THEN 140

120  IF QZ = 0 THEN GOSUB 80:
    PRINT ";:D":
    QZ = 1

130  MAX%#(X) = Y:
    GOTO 180

140  IF Y > = MIN%#(X) THEN 170

```

Listing I (EEM 3-D PLOT)

```

150  IF QZ = 0 THEN  GOSUB 80:
      PRINT ";;D":
      QZ = 1

160  MIN%#(X) = Y:
      GOTO 200

170  IF QZ = 1 THEN  GOSUB 80:
      QZ = 0:
      IF PU = 1 THEN  PRINT ";;U"

180  IF Y < MIN%#(X) THEN MIN%#(X) = Y

190  IF (PD = 0 AND X = X2) OR (PD = 1 AND Y = Y2) THEN 80

200  RETURN

210  PX = XA - .5:
      PY = YA - .5:
      PZ = ZA:
      GOSUB 20

220  XS =  INT (SX + .5):
      YS =  INT (SY + .5)

230  RETURN

240  PX = XB - .5:
      PY = YB - .5:
      PZ = ZB:
      GOSUB 20

250  XE =  INT (SX + .5):
      YE =  INT (SY + .5)

260  IF XS = XE THEN 360

270  SL = (YS - YE) / (XS - XE)

280  B = YS - SL * XS

290  X1 = XS:
      X2 = XE:
      Y1 = YS:
      IF XS > XE THEN X1 = XE:
      X2 = XS:
      Y1 = YE

300  PRINT ";;U":
      X = X1:
      Y = Y1:
      GOSUB 80:
      QZ = 0:
      PD = 0:
      IF PU = 0 THEN  PRINT ";;D"

```

Listing I (EEM 3-D PLOT)

```

310   FOR X = X1 TO X2
320   Y = INT ((SL * X + B) + .5)
330   GOSUB 110
340   NEXT
350   GOTO 420
360   IF YS = YE THEN 420
370   Y1 = YS:
      Y2 = YE:
      X1 = XS:
      IF YS > YE THEN Y1 = YE:
      Y2 = YS:
      X1 = XE
380   PRINT ";:U":
      X = X1:
      Y = Y1:
      GOSUB 80:
      QZ = 0:
      PD = 1:
      IF PU = 0 THEN PRINT ";:D"
390   FOR Y = Y1 TO Y2
400   GOSUB 110
410   NEXT
420   RETURN
427   REM *****
428   REM Hidden line min/max array clearing routine
429   REM *****
430   FOR XX = 1 TO 1969:
      MIN%#(XX) = 1757:
      MAX%#(XX) = 0:
      NEXT
440   RETURN
997   REM *****
998   REM M A I N   P R O G R A M
999   REM *****

```

Listing I (EEM 3-D PLOT)

```

1000  RHO = 10:
      THETA = .9:
      PHI = .9:
      D = 590:
      PU = 1:
      AF = 1

1020  S1 = SIN (TH):
      C1 = COS (TH):
      S2 = SIN (PH):
      C2 = COS (PH)

1040  TEXT :
      HOME :
      PRINT "E E M[3]3 - D[3]P L O T":
      PRINT :
      PRINT

1060  HIMEM: 33019

1080  DIM MAX%#(1969),MIN%#(1969):
      I$ = CHR$ (1)

1100  GOSUB 430

1120  D$ = CHR$ (13) + CHR$ (4)

1140  PRINT "1st Filename= ";:
      CALL - 657

1150  S$ = "":
      FOR WW = 512 TO 767:
      IF PEEK (WW) < > 141 THEN S$ = S$ + CHR$ ( PEEK (WW) - 128)
      :
      NEXT

1160  DIM JP%(72,107)

1180  BA% = PEEK (110) * 256 + PEEK (109) - 15777

1200  PRINT D$"BLOAD"S$,A"BA%",D2"

1220  ML$ = "(x" + STR$ ( INT ((1000 / JP%(13,0) + .05) * 10) / 10)
      + ")"

1240  INPUT "Do you want the time shown? (Y/N) ";TM$:
      PRINT :
      IF TM$ < > "Y" AND TM$ < > "N" THEN 1240

1260  INPUT "Do you want 180 rot.? (Y/N) ";PR$:
      PRINT :
      IF PR$ < > "N" AND PR$ < > "Y" THEN 1260

1280  INPUT "Ratio, Correction or Neither? (R/C/N) ";HC$:
      IF HC$ < > "R" AND HC$ < > "C" AND HC$ < > "N" THEN 1280

```

```

1300 IF HC$ = "N" THEN 1880

1310 PRINT :
    PRINT "2nd Filename= ";:
    CALL - 657

1320 S2$ = "":
    FOR WW = 512 TO 767:
    IF PEEK (WW) < > 141 THEN S2$ = S2$ + CHR$ ( PEEK (WW) - 12
    8):
    NEXT

1340 DIM JP%#(72,107)

1350 FOR X = 0 TO 19:
    JP%#(X,0) = JP%(X,0):
    NEXT

1360 ST = 2:
    IF JP%(0,0) = 5 THEN ST = 1

1380 FOR X = 0 TO 72 STEP ST

1400 FOR M = 1 TO 107 STEP ST

1420 JP%#(X,M) = JP%(X,M)

1440 NEXT M:
    NEXT X

1450 BA% = PEEK (110) * 256 + PEEK (109) - 15777:
    PRINT D$"BLOAD"S2$,A"BA%",D2"

1460 IF HC$ = "R" THEN QL$ = "(x" + STR$ ( INT ((1000 / JP%(13,0)
    + .05) * 10) / 10) + "":
    GOTO 1500

1470 RX = JP%#(13,0):
    PRINT :
    PRINT "Select PEAK: A - (EX270,EM330)":
    PRINT "[13]B - (EX410,EM470)":
    PRINT "[13]C - (EX500,EM560)":
    PRINT "[13]D - yourEx,yourEm":
    GET PK$:
    ON ( ASC (PK$) - 64) GOTO 1480,1485,1490,1491

1475 IF ASC (PK$) < 65 OR ASC (PK$) > 67 THEN 1470

1480 XP = 6:
    MP = 7:
    GOTO 1495

1485 XP = 34:
    MP = 35:
    GOTO 1495

```



```

.....
1490   XP = 52:
      MP = 53:
      GOTO 1495

1491   PRINT :
      INPUT "Enter Ex & Em WL ";XP,MP

1492   XP = INT ((XP - 240) / 5):
      MP = INT ((MP - 295) / 5)

1495   AF = JP%(XP,MP) / JP%(XP,MP)

1500   MX = 0:
      MN = 100

1520   XS = JP%(1,0):
      IF JP%(1,0) > XS THEN XS = JP%(1,0)

1540   XS = (XS - 240) / 5

1560   XE = JP%(2,0):
      IF JP%(2,0) < XE THEN XE = JP%(2,0)

1580   XE = (XE - 240) / 5

1600   MS = JP%(3,0):
      IF JP%(3,0) > MS THEN MS = JP%(3,0)

1620   MS = (MS - 295) / 5

1640   ME = JP%(4,0):
      IF JP%(4,0) < ME THEN ME = JP%(4,0)

1660   ME = (ME - 295) / 5

1680   ST = 2:
      IF JP%(0,0) = 5 AND JP%#(0,0) = 5 THEN ST = 1

1700   FOR X = XS TO XE STEP ST

1720   FOR M = MS TO ME STEP ST

1740   IF JP%(X,M) = 0 AND HC$ = "R" THEN 1840

1760   IF HC$ = "R" THEN JP%(X,M) = (JP%#(X,M) / JP%#(13,0)) / (JP%(X
,M) / JP%(13,0)) * 1000 + .5

1780   IF HC$ = "C" THEN A = JP%#(X,M):
      B = JP%(X,M) * AF:
      JP%(X,M) = (A + 1000) / (B + 1000) * 1000 - 1000

1790   IF JP%(X,M) < 0 THEN JP%(X,M) = 0

1800   IF JP%(X,M) > MX THEN MX = JP%(X,M):
      JP%(14,0) = X:
      JP%(15,0) = M
.....

```

Listing I (EEM 3-D PLOT)

```

1820  IF JP%(X,M) < MN THEN MN = JP%(X,M)

1840  NEXT M:
      NEXT X

1860  JP%(13,0) = MX:
      JP%(16,0) = MN:
      JP%(1,0) = XS * 5 + 240:
      JP%(2,0) = XE * 5 + 240:
      JP%(3,0) = MS * 5 + 295:
      JP%(4,0) = ME * 5 + 295

1880  MX = JP%(13,0):
      OV = JP%(16,0)

1890  IF HC$ = "C" THEN PRINT :
      PRINT S$" original maximum was":
      PRINT RX / 10:
      JP%(5,0) = JP%#(5,0)

1900  PRINT :
      PRINT "Maximum Value (Default= "MX / 10")":
      PRINT "at Ex"5 * JP%(14,0) + 240("JP%(14,0)", Em"5 * JP%(15,
0) + 295("JP%(15,0)");:
      INPUT "[4]";SS$:
      PRINT :
      IF SS$ < > "" THEN MX = VAL (SS$) * 10

1920  PRINT "Offset Value (Default= "OV / 10")";:
      INPUT " ";SS$:
      PRINT :
      IF SS$ < > "" THEN OV = VAL (SS$) * 10

1940  NF = 100 / (MX - OV)

1960  RL$ = "(x" + LEFT$ ( STR$ (NF + .0005),4) + ")"

1980  PRINT :
      INPUT "Plot Small, Large or Cat. page (S/L/C) ";PLT$

2000  DT$ = STR$ (JP%(9,0)) + "/" + STR$ (JP%(10,0)) + "/" + STR$
      (JP%(8,0))

2020  PRINT "Starting Ex (Default= "JP%(1,0));:
      INPUT ") ";SS$:
      PRINT :
      IF SS$ < > "" AND VAL (SS$) < JP%(1,0) THEN 2020

2040  IF SS$ < > "" THEN JP%(1,0) = VAL (SS$)

2060  PRINT "Ending Ex[3](Default= "JP%(2,0));:
      INPUT ") ";SS$:
      PRINT :
      IF SS$ < > "" AND VAL (SS$) > JP%(2,0) THEN 2060

2080  IF SS$ < > "" THEN JP%(2,0) = VAL (SS$)

```

Listing I (EEM 3-D PLOT)

```

2100 PRINT "Starting Em (Default= "JP%(3,0);:
      INPUT ") ";SS$:
      PRINT :
      IF SS$ < > "" AND VAL (SS$) < JP%(3,0) THEN 2100

2120 IF SS$ < > "" THEN JP%(3,0) = VAL (SS$)

2140 PRINT "Ending Em[3](Default= "JP%(4,0);:
      INPUT ") ";SS$:
      PRINT :
      IF SS$ < > "" AND VAL (SS$) > JP%(4,0) THEN 2140

2160 IF SS$ < > "" THEN JP%(4,0) = VAL (SS$)

2180 SS$ = "":
      ZZ = FRE (0)

2200 EL$ = "_ "

2220 DQ% = (JP%(2,0) - JP%(1,0)) / 5

2240 DZ% = (JP%(4,0) - JP%(3,0)) / 5

2260 BL$ = " ;:H A 100,10 S12+ "

2280 PRINT :
      PRINT "BE SURE PLOTTER IS SELECTED![2]THEN PRESS RETURN.":
      GET AZ$:
      PRINT

2300 PRINT D$"PR#1":
      PRINT I$"4D":
      PRINT I$"14B":
      PRINT " ;:P1"

2320 IF PLT$ = "C" THEN PRINT " ;:A W 0,75,2438,1825,0,970,1490,198
      0":
      GOTO 2500

2340 IF PLT$ = "S" THEN PRINT " ;:EH":
      PRINT " ;:ECM":
      PRINT " ;:V40"

2350 IF PLT$ < > "S" THEN PRINT " ;:V16"

2360 IF HC$ = "R" THEN PRINT BL$("S$ML$ / "S2$QL$")";

2380 IF HC$ = "C" THEN PRINT BL$$S$ (vs "S2$") ["PK$"][5]GAIN= " +
      STR$ (JP%(5,0) / 1000);

2400 IF HC$ = "N" THEN PRINT BL$$S$[5]"DT$"[5]GAIN= " + STR$ (JP%
      (5,0) / 1000);

2420 PRINT RL$EL$

2440 PRINT " ;:HAD 0,1825 2438,1825 2438,0 0,0 U"

```

Listing I (EEM 3-D PLOT)

```

2460  IF TM$ = "Y" THEN PRINT ";;H A 50,1750 S12+ "JP%(11,0)":" RIG
      HT$ ("00" + STR$ (JP%(12,0)),2)EL$

2480  XA = 1:
      XB = XA

2500  ZA = 0:
      ZB = ZA

2520  BL$ = "":
      DT$ = "":
      ML$ = ""

2540  YA = 0

2560  YB = .7142857143

2580  GOSUB 210:
      GOSUB 240

2600  PRINT ";;:U"

2620  XS$ = STR$ ( INT ((XS + XE) / 2 + 32.5))

2640  YS$ = STR$ ( INT ((YS + YE) / 2 + 95.5))

2660  BX$ = ";;:A " + XS$ + "," + YS$ + " S(S5,NI,G0,X163,Y-130)"

2680  PRINT BX$"Excitation (nm)"EL$:
      BX$ = ""

2700  GOSUB 430

2720  XB = 1.025

2740  FOR I = 0 TO DQ% STEP 4

2760  YA = (DQ% - I) / DQ% * .7142857143

2780  YB = YA

2800  GOSUB 210:
      GOSUB 240

2820  PRINT ";;:U"

2840  IF (I / 8 - INT (I / 8)) > 1E - 5 THEN 2960

2860  SX$ = STR$ (XE + 230):
      SY$ = STR$ (YE + 70)

2880  N = JP%(1,0) + 5 * I:
      IF PR$ = "Y" THEN N = JP%(2,0) - 5 * I

2900  N$ = STR$ (N)

```

Listing I (EEM 3-D PLOT)

```

2920    NX$ = ";:A " + SX$ + "," + SY$ + " S(S4,NI,G0,X163,Y-130)" + N
      $ + EL$

2940    PRINT NX$

2960    NEXT I:
      REM FOR AT 2740

2980    XB = 0

3000    YA = .7142857143:
      YB = YA

3020    GOSUB 210:
      GOSUB 240

3040    PRINT ";:U"

3060    XS$ = STR$ ( INT ((XS + XE) / 2 + 175.5))

3080    YS$ = STR$ ( INT ((YS + YE) / 2 - 149.5))

3100    BM$ = ";:A " + XS$ + "," + YS$ + " S(S5,NI,G0,X398,Y191)"

3120    PRINT BM$"Emission (nm)"EL$:
      BM$ = ""

3140    GOSUB 430

3160    YB = .7392857143

3180    FOR I = 0 TO DZ% STEP 4

3200    XA = (DZ% - I) / DZ%:
      XB = XA

3220    GOSUB 210:
      GOSUB 240

3240    PRINT ";:U"

3260    IF (I / 8 - INT (I / 8)) > 1E - 5 THEN 3380

3280    SX$ = STR$ (XE + 270):
      SY$ = STR$ (YE + 30)

3300    N = JP%(3,0) + 5 * I:
      IF PR$ = "Y" THEN N = JP%(4,0) - 5 * I

3320    N$ = STR$ (N)

3340    NX$ = ";:A " + SX$ + "," + SY$ + " S(S4,NI,G0,X398,Y191)" + N$
      + EL$

3360    PRINT NX$

```

Listing I (EEM 3-D PLOT)

```

.....

3380  NEXT I:
      REM FOR AT 3180

3400  GOSUB 430

3420  DIM LP#((JP%(2,0) - JP%(1,0)) / JP%(0,0),1)

3440  IF PR$ = "Y" THEN FOR I = JP%(4,0) TO JP%(3,0) STEP - JP%(0,
0):
      GOTO 3480

3460  FOR I = JP%(3,0) TO JP%(4,0) STEP JP%(0,0):
      REM Step X-Emission

3480  EM = (I - 295) / 5

3500  XB = (JP%(4,0) - I) / (JP%(4,0) - JP%(3,0))

3520  IF PR$ = "Y" THEN XB = (I - JP%(3,0)) / (JP%(4,0) - JP%(3,0))

3540  ZB = ZA

3560  XA = XB + JP%(0,0) / (JP%(4,0) - JP%(3,0)):
      IF (PR$ = "N" AND I = JP%(3,0)) OR (PR$ = "Y" AND I = JP%(4,0)
) THEN XA = XB

3580  YA = .7142857143:
      YB = YA

3600  GOSUB 210:
      GOSUB 240

3620  ZB = (JP%((JP%(1,0) - 240) / 5,EM) - OV) * NF / 320

3640  IF PR$ = "Y" THEN ZB = (JP%((JP%(2,0) - 240) / 5,EM) - OV) * N
F / 320

3660  XA = XB:
      CJ = 0:
      PU = 0

3680  GOSUB 210:
      GOSUB 240

3700  PU = 1

3720  IF PR$ = "Y" THEN FOR J = JP%(2,0) TO JP%(1,0) STEP - JP%(0,
0):
      GOTO 3760

3740  FOR J = JP%(1,0) TO JP%(2,0) STEP JP%(0,0):
      REM Step Y-Excitation

3760  EX = (J - 240) / 5

3780  YA = (JP%(2,0) - J) / (JP%(2,0) - JP%(1,0)) * .7142857143
.....

```

Listing I (EEM 3-D PLOT)

```

3800  IF PR$ = "Y" THEN YA = (J - JP%(1,0)) / (JP%(2,0) - JP%(1,0))
      * .7142857143

3820  XS = LP#(CJ,0):
      YS = LP#(CJ,1)

3840  IF (PR$ = "N" AND I = JP%(3,0)) OR (PR$ = "Y" AND I = JP%(4,0))
      ) THEN GOSUB 210

3860  YB = YA

3880  ZB = (JP%(EX,EM) - OV) * NF / 320

3900  GOSUB 240

3920  LP#(CJ,0) = XE:
      LP#(CJ,1) = YE:
      CJ = CJ + 1

3940  NEXT J:
      REM FOR AT 3740

3960  IF (PR$ = "N" AND I = JP%(3,0)) OR (PR$ = "Y" AND I = JP%(4,0))
      ) THEN ZB = ZA:
      YB = .7142857143:
      GOSUB 210:
      GOSUB 240

3980  FOR JJ = CJ - 2 TO 0 STEP - 1

4000  XS = LP#(JJ,0):
      YS = LP#(JJ,1)

4020  XE = LP#(JJ + 1,0):
      YE = LP#(JJ + 1,1)

4040  GOSUB 260

4060  NEXT JJ:
      NEXT I:
      REM FOR AT 3980 (JJ) & 3460 (I)

4080  PRINT D$"PR#1":
      PRINT "":EF":
      PRINT "":HAU 3400,0":
      PRINT "":@":

4100  PRINT CHR$ (1)"R"

4120  PRINT D$"RUN MENU,S6,D1"

```

END Listing I (EEM 3-D PLOT)

Listing II (GW-BASIC EEM 3-D)

```

.....

997  REM *****
998  REM M A I N   P R O G R A M
999  REM *****

1000  RHO = 10:
      THETA = .9:
      PHI = .9:
      D = 590:
      PU = 1:
      AF = 1

1020  S1 = SIN (THETA):
      C1 = COS (THETA):
      S2 = SIN (PHI):
      C2 = COS (PHI)

1040      CLS:
      PRINT "E E M[3]3 - D[3]P L O T":
      PRINT :
      PRINT

1060  DIM MAX%(1969),MIN%(1969)

1080  GOSUB 7000

1100  INPUT "Filename= ";S$

1120  DIM EEM%(47,50)

1140      OPEN S$ FOR INPUT AS #1

1160  INPUT #1, ;STRTEX,ENDEX,STRTEM,ENDEM,MAXIMUM,MINIMUM,MEAN,STD
      EV,DA,MO,YR,HR,MIN

1180  INPUT "Do you want the time shown? (Y/N) ";TM$:
      PRINT :
      IF TM$ < > "Y" AND TM$ < > "N" THEN 1180

1200  INPUT "Do you want 180 rot.? (Y/N) ";PR$:
      PRINT :
      IF PR$ < > "N" AND PR$ < > "Y" THEN 1200

1220  PRINT "Maximum Value (Default= "MAXIMUM")";:
      INPUT " ";SS$:
      PRINT :
      IF SS$ < > "" THEN MAXIMUM = VAL (SS$)

1240  PRINT "Minimum Value (Default= "MINIMUM")";:
      INPUT " ";SS$:
      PRINT :
      IF SS$ < > "" THEN MINIMUM = VAL (SS$)

1260  NF = 100 / (MAXIMUM - MINIMUM)
.....

```



```

.....
1280   FOR XX = 0 TO 47

1300   FOR MM = 0 TO 50

1320   INPUT #1, ;EEM:
      EEM%(XX,MM) = INT ((EEM - MINIMUM) * NF + .5)

1340   IF EEM%(XX,MM) > 100 THEN EEM%(XX,MM) = 100

1360   IF EEM%(XX,MM) < 0 THEN EEM%(XX,MM) = 0

1380   NEXT MM

1400   NEXT XX

1420       CLOSE#1

1440   RL$ = "(x" + STR$ (NF) + ")"

1460   DT$ = STR$ (MO) + "/" + STR$ (DA) + "/" + STR$ (YR)

1480   PRINT "Starting Ex (Default= "STRTEX;:
      INPUT ")";SS$:
      PRINT :
      IF SS$ < > "" AND VAL (SS$) < STRTEX THEN 2020

1500   IF SS$ < > "" THEN STRTEX = VAL (SS$)

1520   PRINT "Ending Ex[3](Default= "ENDEX;:
      INPUT ") ";SS$:
      PRINT :
      IF SS$ < > "" AND VAL (SS$) > ENDEX THEN 1520

1540   IF SS$ < > "" THEN ENDEX = VAL (SS$)

1560   PRINT "Starting Em (Default= "STRTEM;:
      INPUT ") ";SS$:
      PRINT :
      IF SS$ < > "" AND VAL (SS$) < STRTEM THEN 1560

1580   IF SS$ < > "" THEN STRTEM = VAL (SS$)

1600   PRINT "Ending Em[3](Default= "ENDEM;:
      INPUT ") ";SS$:
      PRINT :
      IF SS$ < > "" AND VAL (SS$) > ENDEM THEN 1600

1620   IF SS$ < > "" THEN ENDEM = VAL (SS$)

1640   EL$ = CHR$ (3)

1660   DQ% = (ENDEX - STRTEX) / 10

1680   DZ% = (ENDEM - STRTEM) / 10

1700   BL$ = "PA 720,560 SI .25,.35 LB"
.....

```

```

1720  PRINT :
      PRINT "BE SURE PLOTTER IS SELECTED![2]THEN PRESS RETURN."

1740  AZ$ = INKEY$:
      IF LEN (AZ$) = 0 THEN 1740

1760      OPEN"COM1:1200,E,7,1"AS#1

1780  PRINT #1, "IN SP1 DI;"

1800  PRINT #1, BL$$"[5]"DT$"[5]"RL$EL$

1820  PRINT #1, "PU PA 320,520 PD PA 320,7820 10072,7820 10072,520
      320,520 PU;"

1840  IF TM$ = "Y" THEN PRINT #1, "PA 520,7520 LB" STR$ (HR)":" RI
      GHT$ ("00" + STR$ (MIN),2)EL$

1860  XA = 1:
      XB = XA

1880  ZA = 0:
      ZB = ZA

1900  BL$ = "":
      DT$ = "":
      ML$ = ""

1920  YA = 0:
      YB = .7142857143

1940  GOSUB 5000:
      GOSUB 6000

1960  PRINT #1, "PU;"

1980  XS$ = STR$ ( INT ((XS + XE) * 2 + 130))

2000  YS$ = STR$ ( INT ((YS + YE) * 2 + 382))

2020  BX$ = "PA " + XS$ + "," + YS$ + " DI .7772,-.6293 LB"

2040  PRINT #1, BX$"Excitation (nm)"EL$

2060  GOSUB 7000

2080  XB = 1.025

2100  FOR I = 0 TO DQ% STEP 2

2120  YA = (DQ% - I) / DQ% * .7142857143

2140  YB = YA

2160  GOSUB 5000:
      GOSUB 6000

```

```

2180 PRINT #1, "PU;"

2200 IF (I / 8 - INT (I / 8)) > 1E - 5 THEN 2320

2220 SX$ = STR$ (4 * XE + 780):
      SY$ = STR$ (4 * YE + 320)

2240 N = STRTEX + 10 * I:
      IF PR$ = "Y" THEN N = ENDEX - 10 * I

2260 N$ = STR$ (N)

2280 NX$ = "PA " + SX$ + "," + SY$ + "SI .25,.35 DI .9032,.429 LB"
      + N$ + EL$

2300 PRINT #1, NX$"PU;"

2320 NEXT I

2340 XB = 0

2360 YA = .7142857143:
      YB = YA

2380 GOSUB 5000:
      GOSUB 6000

2400 PRINT #1, "PU;"

2420 XS$ = STR$ ( INT ((XS + XE) * 2 + 702))

2440 YS$ = STR$ ( INT ((YS + YE) * 2 - 598))

2460 BM$ = "PA " + XS$ + "," + YS$ + "SI .25,.35 DI .9032,.429 LB"

2480 PRINT #1, BM$"Emission (nm)"EL$

2500 GOSUB 7000

2520 YB = .7392857143

2540 FOR I = 0 TO DZ% STEP 2

2560 XA = (DZ% - I) / DZ%:
      XB = XA

2580 GOSUB 5000:
      GOSUB 6000

2600 PRINT #1, "PU;"

2620 IF (I / 8 - INT (I / 8)) > 1E - 5 THEN 2740

2640 SX$ = STR$ (4 * XE + 1000):
      SY$ = STR$ (4 * YE + 20)

```

```

.....

2660  N = STRTEM + 10 * I:
      IF PR$ = "Y" THEN N = ENDEM - 10 * I

2680  N$ =  STR$ (N)

2700  NX$ = "PA " + SX$ + "," + SY$ + "SI .2,.3 DI .9032,.429 LB" +
      N$ + EL$

2720  PRINT  #1, NX$"PU;"

2740  NEXT I

2760  GOSUB 7000

2780  DIM LP((ENDEX - STRTEX) / 10,1)

2800  IF PR$ = "Y" THEN 2900

2820  FOR I = STRTEM TO ENDEM STEP 10

2840  GOSUB 9000

2860  NEXT I

2880  GOTO 2960

2900  FOR I = ENDEM TO STRTEM STEP  - 10

2920  GOSUB 9000

2940  NEXT I

2960  PRINT  #1, "SP0"

2980  REM CLOSE #1

2990  END

2997  REM *****
2998  REM 3 - D to 2 - d translation routine
2999  REM *****

3000  TX =  - PX * S1 + PY * C1

3010  TY =  - PX * C1 * C2 - PY * S1 * C2 + PZ * S2

3020  TZ =  - PX * S2 * C1 - PY * S2 * S1 - PZ * C2 + RHO

3030  SX = (D * TX / TZ + 36.6050347) * 24.7680682 + 240

3040  SY = (D * TY / TZ + 25.2546018) * 23.4965782 + 131

3050  RETURN

.....

```

```
.....

3497 REM *****
3498 REM Hewlett Packard line segment plot command
3499 REM *****
3500 PRINT #1,"PA",(X + 280) * 4,"",(Y + 75) * 4,";"
3510 RETURN

3997 REM *****
3998 REM Hidden line removal routine
3999 REM *****

4000 IF Y < = MAX%(X) THEN 4030
4010 IF QZ = 0 THEN GOSUB 3500:
PRINT #1,"PD;":
QZ = 1

4020 MAX%(X) = Y:
GOTO 4070

4030 IF Y > = MIN%(X) THEN 4060
4040 IF QZ = 0 THEN GOSUB 3500:
PRINT #1,"PD;":
QZ = 1

4050 MIN%(X) = Y:
GOTO 4090

4060 IF QZ = 1 THEN GOSUB 3500:
QZ = 0:
IF PU = 1 THEN PRINT #1,"PU;"

4070 IF Y < MIN%(X) THEN MIN%(X) = Y
4080 IF (PD = 0 AND X = X2) OR (PD = 1 AND Y = Y2) THEN 3500
4090 RETURN

5000 PX = XA - .5:
PY = YA - .5:
PZ = ZA:
GOSUB 3000

5010 XS = INT (SX + .5):
YS = INT (SY + .5)

5020 RETURN

.....
```

```
.....

6000    PX = XB - .5:
        PY = YB - .5:
        PZ = ZB:
        GOSUB 3000

6010    XE = INT (SX + .5):
        YE = INT (SY + .5)

6020    IF XS = XE THEN 6120

6030    SL = (YS - YE) / (XS - XE)

6040    B = YS - SL * XS

6050    X1 = XS:
        X2 = XE:
        Y1 = YS:
        IF XS > XE THEN X1 = XE:
        X2 = XS:
        Y1 = YE

6060    PRINT #1, "PU;":
        X = X1:
        Y = Y1:
        GOSUB 3500:
        QZ = 0:
        PD = 0:
        IF PU = 0 THEN PRINT #1, "PD;"

6070    FOR X = X1 TO X2

6080    Y = INT ((SL * X + B) + .5)

6090    GOSUB 4000

6100    NEXT

6110    GOTO 6180

6120    IF YS = YE THEN 6180

6130    Y1 = YS:
        Y2 = YE:
        X1 = XS:
        IF YS > YE THEN Y1 = YE:
        Y2 = YS:
        X1 = XE

6140    PRINT #1, "PU;":
        X = X1:
        Y = Y1:
        GOSUB 3500:
        QZ = 0:
        PD = 1:
        IF PU = 0 THEN PRINT #1, "PD;"

.....
```

```

.....

6150   FOR Y = Y1 TO Y2

6160   GOSUB 4000

6170   NEXT

6180   RETURN

6997   REM *****

6998   REM Hidden line min/max array clearing routine

6999   REM *****

7000   FOR XX = 1 TO 1969:
      MIN%(XX) = 1757:
      MAX%(XX) = 0:
      NEXT

7010   RETURN

7997   REM *****

7998   REM Excitation spectra loop routine

7999   REM *****

8000   XX = (J - 230) / 10

8010   YA = (ENDEX - J) / (ENDEX - STRTEX) * .7142857143

8020   IF PR$ = "Y" THEN YA = (J - STRTEX) / (ENDEX - STRTEX) * .7142
      857143

8030   XS = LP(CJ,0):
      YS = LP(CJ,1)

8040   IF (PR$ = "N" AND I = STRTEM) OR (PR$ = "Y" AND I = ENDEM) THE
      N   GOSUB 5000

8050   YB = YA

8060   ZB = EEM%(XX,MM) / 320

8070   GOSUB 6000

8080   LP(CJ,0) = XE:
      LP(CJ,1) = YE:
      CJ = CJ + 1

8090   RETURN

8997   REM *****

8998   REM Emission spectra loop routine
.....

```

```
.....  
8999  REM *****  
9000  MM = (I - 300) / 10  
9010  XB = (ENDEM - I) / (ENDEM - STRTEM)  
9020  IF PR$ = "Y" THEN XB = (I - STRTEM) / (ENDEM - STRTEM)  
9030  ZB = ZA  
9040  XA = XB + 10 / (ENDEM - STRTEM)  
9050  YA = .7142857143:  
      YB = YA  
9060  GOSUB 5000:  
      GOSUB 6000  
9070  PX = (STRTEX - 230) / 10:  
      QX = (ENDEX - 230) / 10  
9080  ZB = EEM%(PX,MM) / 320  
9090  IF PR$ = "Y" THEN ZB = EEM%(QX,MM) / 320  
9100  XA = XB:  
      CJ = 0:  
      PU = 0  
9110  GOSUB 5000:  
      GOSUB 6000  
9120  PU = 1  
9130  IF PR$ = "Y" THEN 9180  
9140  FOR J = STRTEX TO ENDEX STEP 10  
9150  GOSUB 8000  
9160  NEXT J  
9170  GOTO 9210  
9180  FOR J = ENDEX TO STRTEX STEP - 10  
9190  GOSUB 8000  
9200  NEXT J  
9210  IF (PR$ = "N" AND I = STRTEM) OR (PR$ = "Y" AND I = ENDEM) THE  
      N ZB = ZA:  
      YB = .7142857143:  
      GOSUB 5000:  
      GOSUB 6000  
.....
```


.....

9220 FOR JJ = CJ - 2 TO 0 STEP - 1

9230 XS = LP(JJ,0):
YS = LP(JJ,1)

9240 XE = LP(JJ + 1,0):
YE = LP(JJ + 1,1)

9250 GOSUB 6020

9260 NEXT JJ

9270 RETURN

.....

END Listing II (GW-BASIC EEM 3-D)